Rounding Cubic Roots and the Monthly

Alejandro López-Ortiz^1, Luke O’Connor^2

Introduction

In the past, Maple has been used as an aid to solve mathematical problems from the famous problem section of the American Mathematical Monthly journal. For those unfamiliar with the “Monthly”, as the journal is colloquially known, let it be said that the problems are chosen to be at a level where they would challenge even a bright math senior. Some schools have problem groups, in which students contribute towards finding a solution.

The solution to the following problem, proposed by Seung-Jin Bang, combines discrete and continuous operands, finite and infinite structures, and the assume facilities of Maple. As is often the case, the solution to the problem at hand is in the form of an alternating sequence of computer and human insights, which eventually lead to the solution.

The formal mathematician may wish to verify Maple’s proposed solution using formal methods. With the aid of Maple’s solution such task becomes then of a mechanical nature.

Round Cubic Roots

10212. Proposed by Seung-Jin Bang, Seoul, Korea. Let a(n) be the integer closest to n^{1/3}. Evaluate S = ∞∑ n=1 1 a(n)^4 . Note that

> S=sum(Sum(1/m^4,n=ceil((m-1/2)^3)..ceil((m+1/2)^3)-1),m=1..infinity);

S = ∞∑ m=1 ⌈(m+1/2)3⌉−1 ∑ n=⌈(m−1/2)3⌉ m^{−4}

we let Maple know that m is an integer, and force it to expand the arguments of ceil.

> assume(m,integer);
> lm:=expand((m-1/2)^3):
> um:=expand((m+1/2)^3-1):
> S=sum(sum(1/m^4,n=ceil(lm)..ceil(um)),m=1..infinity);

1Department of Computer Science, University of Waterloo, Waterloo, Ontario, Canada, N2L 3G1, alopezo@barrow.uwaterloo.ca.

2Work performed while at the Department of Computer Science, University of Waterloo. Current address: DSTC Pty Ltd, Level 6, Gehrmann Laboratories, The University of Queensland Q 4072. Australia, oconnor@fit.qut.edu.au.

S = ∞∑ m=1 (⌈ 3 m^2 2 + 3 m 4 − 7 8 ⌉ + 1 − ⌈ − 3 m^2 2 + 3 m 4 − 1/8 ⌉) m^{−4} Since Maple now has assumptions on m, it reminds the user of this fact by printing m~ whenever m appears. For purposes of clarity in this presentation, we have removed all the tildes from the output of Maple commands.

Maple needs a bit more help. Let m = 4k + r and

> alpha:=proc (k,r) subs(m=4*k+r,sum(1,n=ceil(lm)..ceil(um))); end:
> assume(k,integer);
> assume(r,integer);
> S=Sum(’alpha(0,r)’/r^4,r=1..3)+
    Sum(Sum(’alpha(k,r)/(4*k+r)^4’,r=0..3),k=1..infinity);

S = 3∑ r=1 α(0, r) r^4 + ∞∑ k=1 3∑ r=0 α(k, r) (4 k + r)^4

> S=Sum(’alpha(k,0)/(4*k)^4’,k=1..infinity)+
    Sum(’alpha(k,1)/(4*k+1)^4’,k=0..infinity)+
    Sum(’alpha(k,2)/(4*k+2)^4’,k=0..infinity)+
    Sum(’alpha(k,3)/(4*k+3)^4’,k=0..infinity);

S = ∞∑ k=1 α(k, 0) 256 k^4 + ∞∑ k=0 α(k, 1) (4 k + 1)^4 + ∞∑ k=0 α(k, 2) (4 k + 2)^4 + ∞∑ k=0 α(k, 3) (4 k + 3)^4

> simplify(");

S = ∞∑ k=1 48 k^2 + 1 k^4 1 256 + 3 ∞∑ k=0 (4 k + 1)^{−2} + 3 4 ∞∑ k=0 (2 k + 1)^{−2} + 3 ∞∑ k=0 (4 k + 3)^{−2}

Eq. 1

Now Maple is ready to provide us with a better expression.

> S=eval(subs(Sum=sum,op(2,")));

S = π^4 23040 + π^2 8 + 3 Ψ(1, 1/4) 16 + 3 Ψ(1, 3/4) 16 This is a solid step forward. Two questions come to mind: What exactly is the psi function?

and, how can we compute it at 1/4 and 3/4?

The Maple help facility reports that psi is the Polygamma function defined as Ψ(x) = d/dx ln(Γ(x)) = (d/dx Γ(x))/Γ(x), and Ψ(1, x) = d/dx Ψ(x). The Gamma Function

Euler’s Gamma function has several interesting properties on the unit interval. More specifically, we have Euler’s Reflection Formula:

> GAMMA(z)*GAMMA(1-z)=simplify(GAMMA(z)*GAMMA(1-z));

Γ(z)Γ(1 − z) = − π sin(π (−1 + z))

> ReflFrml := ":

We need to verify that the theorem holds for the values of z required, namely, for z = 1/4, 3/4. The function is clearly continuous at those values, and indeed is valid for all nonintegral values of z.

Double differentiation of this formula, plus some manipulation of the equalities derived from it gives us the result required.

> d1:=simplify(subs(z=1/4,diff(ReflFrml,z))):
> d2:= simplify(subs(z=1/4,diff(ReflFrml,z$2))):
> d1:=expand(op(1,d1)^2=op(2,d1)^2):
> d1:=expand(d1/(Pi*sqrt(2))):
> expand((d2-d1)/(Pi*sqrt(2)));

Ψ(1, 1/4) + Ψ(1, 3/4) = 2 π^2

Thus the original summation simplifies to ∞∑ n=1 a(n)^{−4} = π^4/23040 + π^2/2.

We shall verify this result numerically in the next section.

Numerical Verification

A simple verification procedure for the result above is to compute the value of the summation for the first few terms m = 1 . . .k, say for k = 50, and compare it to the exact solution computed, viz.

> evalf(Sum((ceil((m+1/2)^3)-ceil((m-1/2)^3))/m^4,m = 1..50))
  = evalf(Pi^4/23040+Pi^2/2);

4.879625361 = 4.939030027

which is inconclusive. Even for k = 10000, the approximation is somewhat unsatisfactory for such large value of k,

> evalf(Sum((ceil((m+1/2)^3)-ceil((m-1/2)^3))/m^4,m = 1..10000))
  = evalf(Pi^4/23040+Pi^2/2);

4.938730041 = 4.939030027

Since the series is not converging rapidly we need some assurances of whether it actually converges and to what value. We can estimate numerically the value of S with more confidence by computing a bound on the magnitude of the error term, i.e. the tail of the summation, as follows. Clearly ∞∑ m=k+1 ⌈(m + 1/2)^3⌉ − ⌈(m − 1/2)^3⌉ m^4 ≤ ∞∑ m=k+1 (m + 1)^3 − (m − 1)^3 m^4

If we expand the binomials on the right hand side and notice that 1/m^4 is smaller than 1/m^2 for m ≥ 1, we obtain ∞∑ m=k+1 (m + 1)^3 − (m − 1)^3 m^4 = 2 ∞∑ m=k+1 3 m^2 + 1 m^4 ≤ 8 ∞∑ m=k+1 1 m^2 ≤ 8 ∫_∞ k 1 x^2 d x

that is,

> sum((ceil((m+1/2)^3)-ceil((m-1/2)^3))/m^4,m = k+1 .. infinity)
  <= 8*int(1/x^2,x=k..infinity);

∞∑ m=k+1 ⌈(m + 1/2)^3⌉ − ⌈(m − 1/2)^3⌉ m^4 ≤ 8 k Having computed this bound we have shown that

> S = evalf(Sum((ceil((m+1/2)^3)-ceil((m-1/2)^3))/m^4,m = 1..10000))+epsilon;

S = 4.938730041 + ϵ

for 0 < ϵ < 0.0008 = 8/10000; implying that the numerical value of the summation is bounded from above by 4.939530, which is only 0.0005 bigger than the computed exact solution. This provides independent evidence to the fact that the series converges, albeit slowly, to the computed exact solution.

Indeed, from this error estimate and the fact that Eq. 1 is a sum of four series of inverse quadratic order, it is easy to see that the series S converges at a rate such that the number of digits of accuracy is directly proportional to log_{10} k, which matches our experimental computations.

Conclusions

It is interesting to note that since the solution was first found by the authors (1992), the amount of work that can be done with Maple has increased significantly with each release. Particularly useful were the assume facilities, which allow Maple to simplify expressions involving

ceil.

References

[1] Seung-Jing Bang. Problem 10212, American Mathematical Monthly, Volume 9, Number 4, 1992. p.361.